```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the first Weyl algebra $A = A_1(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d,t], polynom=[t]):
```

and the left A-module M finitely presented by the matrix defined by:

```
> R := evalm([[d,t,0]]);
```

$$R := \begin{bmatrix} d & t & 0 \end{bmatrix}$$

The rank of the left A-module M is:

```
> OreRank(R,A);
```

$$2$$

Since R admits a right inverse S defined by

```
> S := RightInverse(R,A);
```

$$S := \begin{bmatrix} t \\ -d \\ 0 \end{bmatrix}$$

M is a stably free left A-module of rank 2, i.e., a free left A-module of rank 2. Using the fact that the direct sum of M and A is isomorphic to $A^{1\times 3}$, and using the Cancellation Theorem, let us compute a basis of M. Let us first compute

```
> X := stackmatrix(R,1-Mult(S,R,A));
```

$$X := \begin{bmatrix} d & t & 0 \\ -d\,t+1 & -t^2 & 0 \\ d^2 & 2+d\,t & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

which defines a left A-isomorphism from the direct sum of A and M onto $A^{1\times 3}$. Moreover, the direct sum of A and M is isomorphic to the left A-module L finitely presented by

```
> P := augment(evalm([[0]]),R);
```

$$P := \begin{bmatrix} 0 & d & t & 0 \end{bmatrix}$$

Similarly, a finite presentation of $A^{1\times 2}$ is given by the matrix defined by

```
> Rp := evalm([[0$2]]);
```

$$Rp := \begin{bmatrix} 0 & 0 \end{bmatrix}$$

and $A^{1\times 3}$ is isomorphic to the left A-module L' which is finitely presented by the matrix defined by

```
> Pp := augment(evalm([[0]]),Rp);
```

$$Pp := \left[\begin{array}{ccc} 0 & 0 & 0 \end{array}\right]$$

Let us check again that the left A-homomorphism from L to L' induced by X is a left A-isomorphism:

```
> TestIso(P,Pp,X,A);
```

$$true$$

We can now apply the algorithm corresponding to the Cancellation Theorem to these matrices:

```
> Q := map(collect,Cancellation(Rp,X,A,"splithom"),[d,t]):
```

to obtain a left A-isomorphism h from M to Aˆ(1 x 2) induced by Q

```
> rowdim(Q); coldim(Q);
```

$$3$$
$$2$$

whose first column is defined by

```
> submatrix(Q,1..3,1..1);
```

$$\left[\begin{array}{c} (-t^4 - t^3)\, d^2 + (-3\,t^2 - 4\,t^3 - t^4)\, d - 3\,t^2 - 2\,t^3 \\ (t^2 + t^3)\, d^3 + (t^3 + 6\,t + 8\,t^2)\, d^2 + (15\,t + 6\,t^2 + 6)\, d + 6 + 6\,t \\ -d^2\, t^3 + (-t^3 - 3\,t^2)\, d - t - 2\,t^2 \end{array}\right]$$

and whose second columns is defined by

```
> submatrix(Q,1..3,2..2);
```

$$\left[\begin{array}{c} (-t^4 - t^2 - 2\,t^3)\, d^3 + (-t - t^4 - 6\,t^2 - 6\,t^3)\, d^2 + (-6\,t^2 - 2\,t + 1 - 3\,t^3)\, d - t^2 + 2 \\ 2 + (t + 2\,t^2 + t^3)\, d^4 + (3 + 10\,t^2 + 12\,t + t^3)\, d^3 + (14 + 24\,t + 7\,t^2)\, d^2 + (10\,t + 12)\, d \\ (-t^3 - t^2)\, d^3 + (-t - 5\,t^2 - t^3)\, d^2 + (-3\,t - 3\,t^2)\, d + 1 - t \end{array}\right]$$

Let us check again that h is a left A-isomorphism:

```
> TestIso(R,Rp,Q,A);
```

$$true$$

Thus, the matrix Q defines an injective parametrization of M, i.e., we have $ker_A(.\mathrm{Q}) = A$ R

```
> SyzygyModule(Q,A);
```

$$\left[\begin{array}{ccc} d & t & 0 \end{array}\right]$$

and Q admits a left inverse T

```
> T := map(collect,LeftInverse(Q,A),[d,t]):
> rowdim(T); coldim(T);
```

$$2$$
$$3$$

whose first two columns are

```
> submatrix(T,1..2,1..2);
```

$$\begin{bmatrix} -2 & (t^2+t^3)\,d^2+(5\,t+7\,t^2+t^3)\,d+4+10\,t+5\,t^2 \\ t & -2\,t^2-t^3-t^3\,d \end{bmatrix}$$

and whose last column is defined by

```
> submatrix(T,1..2,3..3);
```

$$\begin{bmatrix} (t+2\,t^2+t^3)\,d^3+(3+11\,t+9\,t^2+t^3)\,d^2+(11+19\,t+6\,t^2)\,d+7+6\,t \\ (-t^3-t^2)\,d^2+(-t-4\,t^2-t^3)\,d+1-2\,t-2\,t^2 \end{bmatrix}$$

Hence, we get the exact sequence

$$0 \longrightarrow A \xrightarrow{.R} A^{1\times 3} \xrightarrow{.Q} A^{1\times 2} \longrightarrow 0,$$

which shows that M, defined by $coker_A$(.R), is isomorphic to $im_A(.Q) = A^{1\times 2}$. In particular, the residue classes of the rows of the left inverse T of Q define a basis of M.